Problem:
Score : 500 points

Problem Statement

You are given a string S of length N consisting of lowercase English letters.

Process Q queries of the following two types:

Type 1: change the i_q-th character of S to c_q. (Do nothing if the i_q-th character is already c_q.)

Type 2: answer the number of different characters occurring in the substring of S between the l_q-th and r_q-th characters (inclusive).

Constraints

N, Q, i_q, l_q, and r_q are integers.

S is a string consisting of lowercase English letters.

c_q is a lowercase English letter.

1 \leq N \leq 500000

1 \leq Q \leq 20000

|S| = N

1 \leq i_q \leq N

1 \leq l_q \leq r_q \leq N

There is at least one query of type 2 in each testcase.

Input

Input is given from Standard Input in the following format:

N
S
Q
Query_1
\vdots
Query_Q

Here, Query_i in the 4-th through (Q+3)-th lines is one of the following:

1 i_q c_q

2 l_q r_q

Output

For each query of type 2, print a line containing the answer.

Sample Input 1

7
abcdbbd
6
2 3 6
1 5 z
2 1 1
1 4 a
1 7 d
2 1 7

Sample Output 1

3
1
5

In the first query, cdbb contains three kinds of letters: b , c , and d, so we print 3.

In the second query, S is modified to abcdzbd.

In the third query, a contains one kind of letter: a, so we print 1.

In the fourth query, S is modified to abcazbd.

In the fifth query, S does not change and is still abcazbd.

In the sixth query, abcazbd contains five kinds of letters: a, b, c, d, and z, so we print 5.

Solution in Pascal (FPC 2.6.2) (judge verdict: Accepted)

type BITDataType=Longint;
type BinaryIndexedTree=record n:Longint;bit:Array of BITDataType;end;
procedure init(var bit:BinaryIndexedTree;N:Longint);
var i:Longint;
begin
	bit.n:=N;
	setlength(bit.bit,N+1);
	for i:=1 to N do bit.bit[i]:=0;
end;
function sum(var bit:BinaryIndexedTree;i:Longint):BITDataType;
var ans:BITDataType;
begin
	ans:=0;
	while i>0 do begin
		inc(ans,bit.bit[i]);
		dec(i,i and-i);
	end;
	sum:=ans;
end;
procedure add(var bit:BinaryIndexedTree;i:Longint;a:BITDataType);
begin
	while i<=bit.n do begin
		inc(bit.bit[i],a);
		inc(i,i and-i);
	end;
end;

var
	N,Q,i,qt,id,l,r,cnt:Longint;
	S:String;
	c:Char;
	bit:Array['a'..'z']of BinaryIndexedTree;
begin
	readln(N);
	readln(S);
	readln(Q);
	for c:='a' to 'z' do init(bit[c],N);
	for i:=1 to N do add(bit[S[i]],i,1);
	for i:=1 to Q do begin
		read(qt);
		if qt=1 then begin
			read(id);
			read(c);
			read(c);
			add(bit[S[id]],id,-1);
			S[id]:=c;
			add(bit[S[id]],id,1);
		end else begin
			read(l,r);
			cnt:=0;
			for c:='a' to 'z' do if sum(bit[c],r)>sum(bit[c],l-1) then inc(cnt);
			writeln(cnt);
		end;
	end;
end.